Wolfram Koepf
Seelingstr. 21
14059 Berlin
koepf@zib.de

Keynote Lecture: Integralrechnung im Zeitalter der Computeralgebra

1 Zusammenfassung

Der Einsatz computeralgebrafähiger Mathematikprogramme in der Mathematikausbildung kann und sollte die zukünftige Didaktik der Integralrechnung beeinflussen. Beim traditionellen Unterricht wird nur kurz auf das Flächenproblem eingegangen. Dann wird das Integrieren recht schnell auf das Antidifferenzieren zurückgeführt.

Graphische Darstellungen sowie symbolisches und numerisches Rechenvermögen moderner Mathematikprogramme wie DERIVE erschließen hier völlig neue Möglichkeiten:

- Schülerinnen und Schüler können graphische Darstellungen von Integralapproximationen praktisch auf Knopfdruck abrufen,
- die Stammfunktionen vieler elementarer Funktionen lassen sich mittels geeigneter Riemannscher Summen bestimmen,
- Rekursionsformeln für bestimmte Integrale lassen sich automatisch mittels partieller Integration erzeugen.

Diese Methode hat gegenüber der traditionellen Vorgehensweise den Vorteil,

- daß die Schülerinnen und Schüler die Qualität von Integralapproximationen viel besser beurteilen können,
- daß sich auch numerische Integrationstechniken problemlos behandeln lassen,
- daß sich historische Bezüge leichter herstellen lassen,
- daß die Bedeutung des Hauptsatzes wesentlich besser verstanden wird.
- Ferner lassen sich quasi als Nebenprodukt bei der Behandlung von Integrationstechniken beispielsweise Reihenentwicklungen behandeln.

Unter Benutzung von DERIVE wird eine neue didaktische Konzeption des Unterrichts der Integralrechnung mit praktischen Beispielen unterrichtsnah vorgestellt.

2 Differentiation

Zunächst möchte ich den Einsatz von Mathematikprogrammen bei der Behandlung der Differentiation demonstrieren, was in [4] ausführlich beschrieben wurde. Zu diesem Zweck verwende ich Mathematica [8].

Ich möchte zeigen, wie man gemeinsam mit den Schülerinnen und Schülern Mathematica das Differenzieren beibringen kann. Nicht daß dies nötig wäre: Mathematica kann bereits wunderbar differenzieren. Aber diese Vorgehensweise fördert das mathematische Verständnis des behandelten Stoffs.

Die mögliche Definition

```
diff[f_,x_]:=Module[{h},Limit[((f/.x->x+h)-f)/h,h->0]]
```

die eine direkte Übertragung der Definition

$$f'(x) = \lim_{h\to 0} \frac{f(x+h)-f(x)}{h}$$

darstellt, macht keinen Sinn: Zur Berechnung des Grenzwerts werden Ableitungen benötigt (Regel von de l'Hospital).[1] In der Praxis werden Ableitungen durch die Anwendung von *Ableitungsregeln* berechnet. So wollen wir auch Mathematica das Differenzieren beibringen.

Als erstes behandelt man gewöhnlich die Differentiation der Potenzfunktionen.

```
In[1]:= diff[c_,x_]:=0 /; FreeQ[c,x]

In[2]:= diff[x_^n_.,x_]:=n*x^(n-1) /; FreeQ[n,x]
```

Die erste Zeile besagt hierbei, daß Konstanten die Ableitung Null haben sollen, und die zweite Zeile setzt

$$(x^n)' = n\,x^{n-1}\ ,$$

falls n nicht von x abhängt. Dabei liest sich die Mathematica-Syntax so: Die Ableitung von c nach x ist definitionsgemäß 0, falls c nicht von x abhängt; die Ableitung von x^n nach x ist $n\,x^{n-1}$, falls n nicht von x abhängt. Diese Regel soll auch gelten (`n_.`), falls $n = 1$ ist; man beachte, daß dann das Muster x^n nicht mehr zu erkennen ist.

Testen wir, was Mathematica gelernt hat:

```
In[3]:= diff[y,x]

Out[3]= 0
```

Richtig, y hängt ja nicht von x ab!

[1] Es gibt zwar auch einen algorithmischen Zugang zur Grenzwertberechnung [3], welcher völlig unabhängig vom Differenzieren ist. Dieser ist aber nicht in Mathematica (wohl aber in Maple) eingebaut.

```
In[4]:= diff[x^(1/2),x]

                1
Out[4]= ---------
        2 Sqrt[x]
```

Da wir vom Exponenten nicht verlangt hatten, daß er ganzzahlig ist, können nun also bereits beliebige Potenzen abgeleitet werden. Aber beim Beispiel

```
In[5]:= diff[2*x,x]

Out[5]= diff[2 x, x]
```

liefert Mathematica als Ausgabe die Eingabe.
Dem kann aber leicht durch die Regel

```
In[6]:= diff[c_*f_,x_]:=c*diff[f,x] /; FreeQ[c,x]
```

abgeholfen werden. Nun kann Mathematica das Problem von eben lösen:

```
In[7]:= diff[2*x,x]

Out[7]= 2
```

und hat überhaupt gelernt, daß man Konstanten vorziehen kann. Allerdings scheitert es an

```
In[8]:= diff[x+x^2,x]

                    2
Out[8]= diff[x + x , x]
```

da wir noch nicht erklärt haben, wie mit einer Summe verfahren werden soll.
Da die Differentiation linear ist, erklären wir also

```
In[9]:= diff[f_+g_,x_]:=diff[f,x]+diff[g,x]
```

Dies hat das Resultat, daß nun

```
In[10]:= diff[x+x^2,x]

Out[10]= 1 + 2 x
```

vereinfacht wird. Obwohl wir die Additivität nur für 2 Summanden definiert haben, wendet Mathematica diese Regel nun auch auf mehrfache Summen an, da Mathematica weiß, daß die Addition eine assoziative und kommutative Operation ist:

```
In[11]:= diff[Sum[k x^k,{k,0,10}],x]

                         2        3        4        5        6
Out[11]= 1 + 4 x + 9 x  + 16 x  + 25 x  + 36 x  + 49 x

                7        8         9
>         + 64 x  + 81 x  + 100 x
```

Als nächstes wenden wir uns nun weiteren Ableitungsregeln zu. Bislang können z. B. keine Produkte abgeleitet werden

```
In[12]:= diff[(2x+x^2)(5+x^2-4x^3),x]

                         2         2       3
Out[12]= diff[(2 x + x ) (5 + x  - 4 x ), x]
```

Daher teilen wir Mathematica die Produktregel mit

```
In[13]:= diff[f_*g_,x_]:=diff[f,x]*g+diff[g,x]*f
```

und erhalten

```
In[14]:= diff[(2x+x^2)(5+x^2-4x^3),x]

                       2           2
Out[14]= (2 x - 12 x ) (2 x + x ) + (2 + 2 x)

                2      3
          (5 + x  - 4 x )
```

Ebenso implementieren wir nun die Quotientenregel.

```
In[15]:= diff[f_./g_,x_]:=(diff[f,x]*g-diff[g,x]*f)/g^2
```

Nun kann die Prozedur `diff` alle rationalen Funktionen ableiten.

```
In[16]:= diff[Sum[x^k,{k,1,7}]/Sum[k x^k,{k,1,5}],x]

                      2      3      4      5
Out[16]= ((x + 2 x  + 3 x  + 4 x  + 5 x )

                         2      3      4      5      6
>        (1 + 2 x + 3 x  + 4 x  + 5 x  + 6 x  + 7 x ) -
```

```
                       2       3       4
>       (1 + 4 x + 9 x  + 16 x  + 25 x )

              2    3    4    5    6    7
>        (x + x  + x  + x  + x  + x  + x )) /

             2      3      4      5 2
>    (x + 2 x  + 3 x  + 4 x  + 5 x )
```

Bevor wir Mathematica nun noch die Kettenregel mitteilen, wollen wir die Ableitungen der elementaren Funktionen erklären:

```
In[17]:= diff[Log[x_],x_]:=1/x
In[18]:= diff[Sin[x_],x_]:=Cos[x]
In[19]:= diff[Cos[x_],x_]:=-Sin[x]
In[20]:= diff[Tan[x_],x_]:=1/Cos[x]^2
In[21]:= diff[Cot[x_],x_]:=-1/Sin[x]^2
In[22]:= diff[ArcSin[x_],x_]:=1/Sqrt[1-x^2]
In[23]:= diff[ArcCos[x_],x_]:=-diff[ArcSin[x],x]
In[24]:= diff[ArcTan[x_],x_]:=1/(1+x^2)
In[25]:= diff[ArcCot[x_],x_]:=-diff[ArcTan[x],x]
```

Schließlich formulieren wir die Kettenregel. Bislang können ja z. B. die Ableitungen

```
In[26]:= diff[Sin[Cos[x]],x]

Out[26]= diff[Sin[Cos[x]], x]

In[27]:= diff[(1+x)^n,x]

                      n
Out[27]= diff[(1 + x) , x]
```

nicht gefunden werden. Durch die beiden Regeln

```
In[28]:= diff[f_[g_],x_]:=diff[f[g],g]*diff[g,x]

In[29]:= diff[f_^g_,x_]:=f^g*diff[g*Log[f],x]
```

wird die Kettenregel implementiert, und unsere obigen Beispiele liefern nun die Resultate

```
In[30]:= diff[Sin[Cos[x]],x]
```

```
Out[30]= -(Cos[Cos[x]] Sin[x])

In[31]:= diff[(1+x)^n,x]

                  -1 + n
Out[31]= n (1 + x)
```

Überhaupt kann Mathematica nun alle durch rationale sowie algebraische Operationen aus den elementaren Funktionen erzeugten Funktionen differenzieren. Nur als Beispiel nehmen wir die folgende Phantasiefunktion

```
In[32]:= diff[ArcCot[Exp[Sqrt[Sum[x^k,{k,1,4}]]
         -Cot[1/x^2]]*Sin[x]],x]

                         2    3    4          -2
              Sqrt[x + x  + x  + x ] - Cot[x  ]
Out[32]= -((E

                               2    3    4          -2
                    Sqrt[x + x  + x  + x ] - Cot[x  ]
>         Cos[x] + E

                          2      3             -2 2
            1 + 2 x + 3 x  + 4 x      2 Csc[x  ]
>         (----------------------- - -----------)
                        2    3    4         3
           2 Sqrt[x + x  + x  + x ]        x

>         Sin[x]) /

                          2    3    4            -2
             2 Sqrt[x + x  + x  + x ] - 2 Cot[x  ]
>      (1 + E

                 2
>         Sin[x] ))
```

Ich möchte betonen, daß dieser Kursus kein Ersatz für den normalen Unterricht sein soll. Ganz im Gegenteil: Erst wenn die Schülerinnen und Schüler die Differentiation durchgenommen haben, kann durch das vorgestellte Projekt der Stoff vertieft werden. Ich bin überzeugt, daß Schülerinnen und Schüler, welche dieses Projekt erfolgreich durchgeführt haben, das Wesen der Differentiation verstanden haben.

3 Integration

Während das Differenzieren nach Erstellung dieses Programms (bzw., nachdem die Schülerinnen und Schüler die Technik des Differenzierens verstanden haben) *trivialisiert* ist, liegt beim Integrieren eine viel kompliziertere Situation vor.

Beim traditionellen Unterricht wird nur kurz auf das Flächenproblem eingegangen. Dann wird das Integrieren mit Bezug auf den Hauptsatz der Differential- und Integralrechnung recht schnell auf die *Antidifferentiation* zurückgeführt.

Graphische Darstellungen sowie symbolisches und numerisches Rechenvermögen moderner Mathematikprogramme wie DERIVE erschließen hier völlig neue Möglichkeiten. Wir folgen hier den Ausführungen in [7], Kapitel 5, wo DERIVE verwendet wurde. Hierfür kann im Prinzip natürlich jedes Computeralgebrasystem verwendet werden, und auch der TI-92 kann zum Einsatz kommen.

Es lassen sich einfache DERIVE-Prozeduren erklären für die Berechnung Riemannscher Summen und ihrer Grenzwerte bei einer arithmetischen bzw. geometrischen Zerlegung.

Mit arithmetischer Zerlegung kann dann jede Schülerin und jeder Schüler beispielsweise die folgenden Ergebnisse für die Grenzwerte der linken Riemannschen Summen

$$\int_a^b f(x)\,dx = \lim_{n\to\infty} \frac{b-a}{n} \sum_{k=1}^{n} f\left(a+(k-1)\,\frac{b-a}{n}\right)$$

erzielen:

$f(x)$	DERIVE Ausgabe
x^3	$\frac{b^4}{4}-\frac{a^4}{4}$
e^x	e^b-e^a
$\sin x$	$\cos a-\cos b$
$\cos x$	$\sin b-\sin a$

Mit einer geometrischen Zerlegung bekommt man auf analoge Weise zusätzlich:

x^m	$\frac{b\,b^m-a\,a^m}{m+1}$
$1/x$	$-\ln\frac{a}{b}$
$\ln x$	$-a\ln a+b\ln b+a-b$

Man beachte, daß nach diesen Versuchen, die alle Schülerinnen und Schüler selbst durchführen können, diese das Muster bei der bestimmten Integration

$$\int_a^b f(x)\,dx = F(b)-F(a)$$

mit einer Integralfunktion F erkennen werden.

Durch Probieren weiterer Beispiele wird aber auch jeder Schülerin und jedem Schüler klar, daß sich viele Funktionen auf diese Weise nicht erfolgreich behandeln lassen.

Ferner können mit einem Computeralgebrasystem ohne weiteres numerische Rechnungen durchgeführt werden. Mit einfachen DERIVE-Prozeduren ist es möglich, die Trapez- und Simpsonregel durchzuführen ([7], Kapitel 5).

Numerische Berechnungen geben Aufschluß über die Güte numerischer Integrationsverfahren:

$$\text{Approximation von } \pi \approx 4 \int_0^1 \sqrt{1-x^2}\, dx$$

n	4	8	16	32	64
Trapez	2.99570	3.08981	3.12325	3.13510	3.13929
Simpson	3.08359	3.12118	3.13439	3.13905	3.14069

$$\text{Approximation von } 2 = \int_0^\pi \sin x\, dx$$

n	4	8	16	32	64
Trapez	1.89611	1.97423	1.99357	1.99839	1.99959
Simpson	2.00455	2.00026	2	2	2

Man kann schön beobachten, wie langsam die Konvergenz beim ersten Integral ist, während das zweite Integral nach sehr wenigen Schritten recht gut approximiert wird. Auch die Überlegenheit des Simpsonverfahrens wird deutlich.



Abbildung 1: Approximationen zur Flächenberechnung des Kreises

Zur weiteren Beurteilung der Güte von Approximationsverfahren eignen sich graphische Darstellungen, die sich ebenfalls sehr leicht mittels geeigneter DERIVE-Prozeduren erzeugen lassen ([7], Kapitel 5).

In Abbildung 1, auf der Approximationen mit einer arithmetischen Zerlegung, mit dem Trapezverfahren sowie mit dem Simpsonverfahren dargestellt sind, sieht man sehr schön, daß für die numerische Integration solche Stellen offenbar besonders schlecht geeignet sind, an denen die Steigung des Integranden besonders groß wird, und am schlimmsten ist die Situation, wenn wie in unserem Beispiel eine vertikale Tangente vorkommt. Nicht einmal das Simpsonverfahren kann einen solchen Kurvenverlauf korrekt modellieren, da ja quadratische Funktionen bekanntlich niemals vertikale Tangenten haben. In solchen Fällen müssen Programme wie DERIVE zur numerischen Approximation gegebenenfalls andere Verfahren anwenden.

Der vorgestellte Zugang zur Integration hat gegenüber der traditionellen Vorgehensweise die folgenden Vorteile:

- Alle Schülerinnen und Schüler können mit DERIVEs Hilfe die Stammfunktionen vieler elementarer Funktionen mittels Riemannscher Summen bestimmen.
- Dieser Zugang orientiert sich mehr an der *historischen Entwicklung*.
- Die Schülerinnen und Schüler können die Bedeutung des *Hauptsatzes* viel besser würdigen, da sie selbst *erleben*, wie schwierig es (selbst mit DERIVE) ist, manche Stammfunktionen zu bestimmen.

4 Integrationstechniken

DERIVEs Fähigkeiten, Integrale zu bestimmen, sind beeindruckend. Die Integrale, die in der Formelsammlung von Bronstein und Semendjajew [1] geschlossen dargestellt werden, werden von DERIVE *allesamt* generiert!

In dieser Sammlung sind aber auch viele Rekursionsformeln für Integrale verzeichnet wie z. B. Integral Nr. 450

$$\int x^n\, e^{ax}\, dx = \frac{1}{a}\, x^n\, e^{ax} - \frac{n}{a} \int x^{n-1}\, e^{ax}\, dx\ .$$

Eine rein mechanische Erzeugung solcher Rekursionsformeln mit DERIVE ist nicht möglich.

In [7], Kapitel 6, wird vorgeführt, wie man die partielle Integration mit DERIVE automatisieren kann, und wie man hiermit solche Rekursionen dennoch mit DERIVE bestimmen kann.

Eine weitere Anwendung der Integration ist die Bestimmung von *Reihenentwicklungen* ([7], Kapitel 7).

Durch Beobachten von Ergebnissen, die man mit DERIVE erzielen kann, können Schülerinnen und Schüler mathematisches Wissen *entdecken*, welches dann im Nachhinein bewiesen werden kann.

Man beobachte die Integrale

$$\int \tan^n x\, dx$$

für $n = 0, \dots, 10$. Man stellt schnell fest, daß die Ergebnisse für gerade und ungerade n erheblich voneinander abweichen.

Die beobachteten Integrale führen zu den Formeln

$$(-1)^n \int_0^x \tan^{2n} t\, dt = \arctan y - y + \frac{y^3}{3} - \frac{y^5}{5} \pm \cdots + \frac{(-1)^n}{2n-1} y^{2n-1}$$

und

$$(-1)^{n+1} \int_0^x \tan^{2n+1} t\, dt = \ln\cos\arctan y + \frac{y^2}{2} - \frac{y^4}{4} \pm \cdots + \frac{(-1)^{n+1}}{2n} y^{2n} \; .$$

welche durch Substitution auch leicht bewiesen werden können.

Hierbei haben wir gesetzt

$$y = \tan x \qquad \text{bzw.} \qquad x = \arctan y \; .$$

Wir erhalten also für $n \to \infty$ die Formeln

$$0 = \arctan y - y + \frac{y^3}{3} - \frac{y^5}{5} \pm \cdots + \frac{(-1)^n}{2n-1} y^{2n-1} + \cdots$$

sowie

$$0 = \ln\cos\arctan y + \frac{y^2}{2} - \frac{y^4}{4} \pm \cdots + \frac{(-1)^{n+1}}{2n} y^{2n} + \cdots$$

oder auch für $y \in [0, 1)$

$$\arctan y = y - \frac{y^3}{3} + \frac{y^5}{5} - \frac{y^7}{7} \pm \cdots$$

sowie

$$-\ln\cos\arctan y = \frac{1}{2} \ln\left(1 + y^2\right) = \frac{y^2}{2} - \frac{y^4}{4} + \frac{y^6}{6} \mp \cdots$$

Damit hat man die *Arkustangensreihe*

$$\arctan y = y - \frac{y^3}{3} + \frac{y^5}{5} - \frac{y^7}{7} \pm \cdots$$

sowie die *Logarithmusreihe*

$$\ln\left(1 + z\right) = z - \frac{z^2}{2} + \frac{z^3}{3} \mp \cdots$$

gefunden. Ein Grenzübergang liefert auch die *Leibnizreihe*

$$1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} \pm \cdots = \arctan 1 = \frac{\pi}{4}$$

sowie die alternierende harmonische Reihe

$$1 - \frac{1}{2} + \frac{1}{3} - \frac{1}{4} \pm \cdots = \ln 2 \; .$$

Die Identität

$$e^x \left(1 - \int\limits_0^x \frac{t^n}{n!}\, e^{-t}\, dt \right) = 1 + x + \frac{x^2}{2} + \cdots + \frac{x^n}{n!} \; ,$$

die ebenfalls mit DERIVE erraten werden kann, und deren Beweis mit dem Hauptsatz einfach ist, führt zu der *Exponentialreihe*

$$e^x = 1 + x + \frac{x^2}{2} + \frac{x^3}{3!} + \ldots \; ,$$

falls der Integrand für $n \to \infty$ gegen Null strebt:

$$\lim_{n\to\infty} \frac{t^n}{n!}\, e^{-t} = 0 \; .$$

Dies folgt aus der *Stirlingschen Formel*

$$n! \sim \sqrt{2\pi n} \left(\frac{n}{e} \right)^n \; .$$

Literatur

[1] Bronstein, I.N. und Semedjajew, K.A., *Taschenbuch der Mathematik*, Verlag Harri Deutsch, Thun und Frankfurt (Main), 1981.

[2] DERIVE User Manual, Version 3.

[3] Gruntz, D.: *On Computing Limits in a Symbolic Manipulation System.* Diss. ETH No. 11432, 1996.

[4] Koepf, W.: Eine Vorstellung von Mathematica und Bemerkungen zur Technik des Differenzierens. Didaktik der Mathematik **21**, 1993, 125–139.

[5] Koepf, W., Ben-Israel, A., Gilbert, R.P.: *Mathematik mit DERIVE.* Verlag Vieweg, Braunschweig/Wiesbaden, 1993.

[6] Koepf, W.: *Höhere Analysis mit DERIVE*, Verlag Vieweg, Braunschweig/Wiesbaden, 1994.

[7] Koepf, W.: *DERIVE für den Mathematikunterricht.* Verlag Vieweg, Braunschweig/Wiesbaden, 1996.

[8] Wolfram, St.: *The Mathematica Book.* Third Edition, Cambridge University Press, 1996.